## Two-dimensional linear geometric noise model

In[1]:=
```
$Assumptions = a ≠ -d && b ≠ 0;
```

In[2]:=
```
(* System *)
A = {{a, b}, {c, d}}; B = {e, f}; S = {{p, 0}, {r, s}};
F[x_] := -A.(x - B)
G[x_] := DiagonalMatrix[x].S
```

In[5]:=
```
(* Moment equations *)
expr =
    D[x[t]^i y[t]^j, {{x[t], y[t]}}].F[{x[t], y[t]}] +
    1/2 Tr[G[{x[t], y[t]}]^T.D[x[t]^i y[t]^j, {{x[t], y[t]}, 2}].G[{x[t], y[t]}]];
mex[i_, j_] := Evaluate[Expand[expr] /.
    Flatten[Table[x[t]^(i+p) y[t]^(j+q) → m_{i+p,j+q}[t], {p, -2, 4}, {q, -2, 4}]]]
```

In[7]:=
```
meq[i_, j_] := m_{i,j}'[t] == FullSimplify[mex[i, j]]
```

## First order necessarily satisfied equation

In[8]:=
```
m01sol = Solve[meq[1, 0], m_{0,1}[t]][[1]] /. m_{0,0}[t] → 1
```

Out[8]=
$$\left\{m_{0,1}[t] \to \frac{a\,e + b\,f - a\,m_{1,0}[t] - m_{1,0}'[t]}{b}\right\}$$

In[9]:=
```
eq1 = FullSimplify[
  MultiplySides[meq[0, 1], b] /. m01sol /. D[m01sol, t] /. m_{0,0}[t] → 1 /. m_{0,0}'[t] → 0]
```

Out[9]=
$$b\,c\,e + (-b\,c + a\,d)\,m_{1,0}[t] + (a + d)\,m_{1,0}'[t] + m_{1,0}''[t] == a\,d\,e$$

## Second order necessarily satisfied equation

In[10]:=
```
m02eq = meq[0, 2]
```

Out[10]=
$$m_{0,2}'[t] == 2\,(c\,e + d\,f)\,m_{0,1}[t] + \left(-2\,d + r^2 + s^2\right)\,m_{0,2}[t] - 2\,c\,m_{1,1}[t]$$

In[11]:=
```
(* Solve for one column in terms of the previous *)
solmij[i_, j_] := Solve[meq[i + 1, j - 1], m_{i,j}[t]][[1]]
solcolj[j_] := Flatten[Table[solmij[i, j], {i, 0, 5}]]
```

In[13]:=
```
(* Substitute recursively *)
eq2 = FullSimplify[
    m02eq /. solcolj[2] /. D[solcolj[2], t]
          /. solcolj[1] /. D[solcolj[1], t] /. D[solcolj[1], {t, 2}]
      /. m0,0[t] → 1 /. Table[D[m0,0[t], {t, i}] → 0, {i, 1, 5}]];
ex2 = eq2[[1]] - eq2[[2]]
```

Out[14]=

$$\frac{1}{2\,b^2}\left(-4\,b\,(c\,e+d\,f)\,\left(a\,e+b\,f-a\,m_{1,0}[t]-m_{1,0}'[t]\right)+\right.$$
$$2\left(-2\,a^2\,e+b\,c\,e-a\,d\,e-2\,a\,b\,f+a\,e\,p\,r+b\,f\,p\,r\right)m_{1,0}'[t]+$$
$$2\,b\,c\left(2\,(a\,e+b\,f)\,m_{1,0}[t]+\left(-2\,a+p^2\right)m_{2,0}[t]-m_{2,0}'[t]\right)+$$
$$\left(-2\,b\,c+\left(2\,a-p^2\right)\,(a+d-p\,r)\right)m_{2,0}'[t]-4\,a\,e\,m_{1,0}''[t]-$$
$$4\,b\,f\,m_{1,0}''[t]+3\,a\,m_{2,0}''[t]+d\,m_{2,0}''[t]-p^2\,m_{2,0}''[t]-p\,r\,m_{2,0}''[t]+$$
$$\left(2\,d-r^2-s^2\right)\left(2\,(a\,e+b\,f)^2+2\left(-2\,a^2\,e+b\,c\,e-a\,d\,e-2\,a\,b\,f+a\,e\,p\,r+b\,f\,p\,r\right)m_{1,0}[t]+\right.$$
$$\left(-2\,b\,c+\left(2\,a-p^2\right)\,(a+d-p\,r)\right)m_{2,0}[t]-4\,(a\,e+b\,f)\,m_{1,0}'[t]+$$
$$\left.\left.(3\,a+d-p\,(p+r))\,m_{2,0}'[t]+m_{2,0}''[t]\right)+m_{2,0}^{(3)}[t]\right)$$

In[15]:=
```
vars = Quiet[Select[Variables[ex2], #[[0]][[1]] === m || #[[0]][[1]][[1]] === m &]]
```

Out[15]=

$$\left\{m_{1,0}[t],\,m_{1,0}'[t],\,m_{2,0}[t],\,m_{2,0}'[t],\,m_{1,0}''[t],\,m_{2,0}''[t],\,m_{2,0}^{(3)}[t]\right\}$$

In[16]:=
```
FullSimplify[Collect[2 b^2 ex2, vars]]
```

Out[16]=

$$-4\,a\,b\,e\,(c\,e+d\,f)-4\,b^2\,f\,(c\,e+d\,f)+$$
$$2\,(a\,e+b\,f)^2\left(2\,d-r^2-s^2\right)+\left(4\,b\,c\,(a\,e+b\,f)+4\,a\,b\,(c\,e+d\,f)+\right.$$
$$\left.2\left(-2\,a^2\,e+b\,c\,e-a\,d\,e-2\,a\,b\,f+a\,e\,p\,r+b\,f\,p\,r\right)\left(2\,d-r^2-s^2\right)\right)m_{1,0}[t]+$$
$$\left(2\,b\,c\left(-2\,a+p^2\right)+\left(-2\,b\,c+\left(2\,a-p^2\right)\,(a+d-p\,r)\right)\left(2\,d-r^2-s^2\right)\right)m_{2,0}[t]+$$
$$2\left(-2\,a^2\,e+3\,b\,c\,e+a\left(-5\,d\,e-2\,b\,f+e\,r\,(p+2\,r)+2\,e\,s^2\right)+b\,f\left(-2\,d+p\,r+2\left(r^2+s^2\right)\right)\right)$$
$$m_{1,0}'[t]+\left(-4\,b\,c+\left(2\,a-p^2\right)\,(a+d-p\,r)+(3\,a+d-p\,(p+r))\left(2\,d-r^2-s^2\right)\right)m_{2,0}'[t]-$$
$$4\,(a\,e+b\,f)\,m_{1,0}''[t]+\left(3\,a+3\,d-p^2-p\,r-r^2-s^2\right)m_{2,0}''[t]+m_{2,0}^{(3)}[t]$$